

Issue XXIII: Category Theory

Максим Сохацький¹

¹ Національний технічний університет України
Київський політехнічний інститут імені Ігоря Сікорського
23 травня 2025 р.

Анотація

Formal definition of Category.

Keywords: Category Theory

Зміст

1 Category Theory	1
1.1 Category	2
1.2 Pullback	4
1.3 Functor	5
1.4 Terminals	6
1.5 Natural Transformation	7
1.6 Adjunction	8
1.7 Modification	9
1.8 The Logic of Cosmos	10

1 Category Theory

Category Theory provides a rigorous framework for abstracting and unifying mathematical structures. Developed in the 1940s by Samuel Eilenberg and Saunders Mac Lane to address coherence problems in algebraic topology, it generalizes relationships between mathematical objects across diverse fields like algebra, geometry, and computer science. Category Theory captures objects and their morphisms—functions preserving structure—as a universal systems theory, akin to a universal algebra of functions, emphasizing composition and transformation. Interpreted as a foundational language, a tool for structural analysis, or a bridge to computer-aided formalization, it solves problems of abstraction and generalization. Categories serve as a stepping stone to topos theory, which enriches logical and geometric insights, and higher cohesive topos theory, extending to infinity-categories for advanced applications.

1.1 Category

First of all very simple category theory up to pullbacks is provided. We give here all definitions only to keep the context valid.

A **category** \mathcal{C} consists of:

- A class of **objects**, $\text{Ob}(\mathcal{C})$,
- A class of **morphisms**, $\text{Hom}_{\mathcal{C}}(X, Y)$, for each pair $X, Y \in \text{Ob}(\mathcal{C})$,
- Composition maps $\circ : \text{Hom}(Y, Z) \times \text{Hom}(X, Y) \rightarrow \text{Hom}(X, Z)$,
- Identity morphisms $\text{id}_X \in \text{Hom}(X, X)$ for each X ,

satisfying associativity and identity laws.

Definition 1. (Category Signature). The signature of category is a $\sum_{A:U} A \rightarrow A \rightarrow U$ where U could be any universe. The pr_1 projection is called Ob and pr_2 projection is called $\text{Hom}(a, b)$, where $a, b : \text{Ob}$.

`cat : U = (A: U) * (A -> A -> U)`

Definition 2. (Precategory). More formal, precategory C consists of the following. (i) A type Ob_C , whose elements are called objects; (ii) for each $a, b : \text{Ob}_C$, a set $\text{Hom}_C(a, b)$, whose elements are called arrows or morphisms. (iii) For each $a : \text{Ob}_C$, a morphism $1_a : \text{Hom}_C(a, a)$, called the identity morphism. (iv) For each $a, b, c : \text{Ob}_C$, a function $\text{Hom}_C(b, c) \rightarrow \text{Hom}_C(a, b) \rightarrow \text{Hom}_C(a, c)$ called composition, and denoted $g \circ f$. (v) For each $a, b : \text{Ob}_C$ and $f : \text{Hom}_C(a, b)$, $f = 1_b \circ f$ and $f = f \circ 1_a$. (vi) For each $a, b, c, d : A$ and $f : \text{Hom}_C(a, b)$, $g : \text{Hom}_C(b, c)$, $h : \text{Hom}_C(c, d)$, $h \circ (g \circ f) = (h \circ g) \circ f$.

```
def cat : U1
:= Σ (ob: U) (hom: ob -> ob -> U), unit

def isPrecategory (C: cat) : U := Σ
  (id: Π (x: C.ob), C.hom x x)
  (o: Π (x y z: C.ob),
   C.hom x y -> C.hom y z -> C.hom x z)
  (homSet: Π (x y: C.ob), isSet (C.hom x y))
  (o-left: Π (x y: C.ob) (f: C.hom x y),
   = (C.hom x y) (o x x y (id x) f) f)
  (o-right: Π (x y: C.ob) (f: C.hom x y),
   = (C.hom x y) (o x y y f (id y)) f)
  (o-assoc: Π (x y z w: C.ob) (f: C.hom x y)
   (g: C.hom y z) (h: C.hom z w),
   = (C.hom x w) (o x z w (o x y z f g) h)
   (o x y w f (o y z w g h))), 1
def precategory: U1 := Σ (C: cat) (P: isPrecategory C), unit
```

Univalent Categories:

```
def isoCat (P: precategory) (A B: P.C.ob) : U := Σ
  (f: P.C.hom A B)
  (g: P.C.hom B A)
  (retract: Path (P.C.hom A A) (P.P.○ A B A f g) (P.P.id A))
  (section: Path (P.C.hom B B) (P.P.○ B A B g f) (P.P.id B)), 1

def isCategory (P: precategory): U
:= Σ (A: P.C.ob), isContr (Π (B: P.C.ob), isoCat P A B)

def category: U1
:= Σ (P: precategory), isCategory P
```

1.2 Pullback

Definition 3. (Categorical Pullback). The pullback of the cospan $A \xrightarrow{f} C \xleftarrow{g} B$ is a object $A \times_C B$ with morphisms $pb_1 : A \times_C B \rightarrow A$, $pb_2 : A \times_C B \rightarrow B$, such that diagram commutes:

$$\begin{array}{ccc} A \times_C B & \xrightarrow{pb_2} & B \\ f \downarrow & \searrow & \downarrow pb_1 \\ A & \xrightarrow{g} & C \end{array}$$

Pullback (\times_C, pb_1, pb_2) must be universal, means for any (D, q_1, q_2) for which diagram also commutes there must exists a unique $u : D \rightarrow \times_C$, such that $pb_1 \circ u = q_1$ and $pb_2 \circ q_2$.

```

def homTo (P: precategory) (X: P.C.ob): U
:= Σ (Y: P.C.ob), P.C.hom Y X

def cospan (P: precategory): U
:= Σ (X: P.C.ob) (_: homTo P X), homTo P X

def hasCospanCone (P: precategory) (D: cospan P) (w: P.C.ob) : U
:= Σ (f: P.C.hom w D.2.1.1) (g: P.C.hom w D.2.2.1),
= (P.C.hom w D.1) (P.P.○ w D.2.1.1 D.1 f D.2.1.2)
(P.P.○ w D.2.2.1 D.1 g D.2.2.2)
def cospanCone (P: precategory) (D: cospan P): U
:= Σ (w: P.C.ob), hasCospanCone P D w

def isCospanConeHom (P: precategory) (D: cospan P)
(E1 E2: cospanCone P D) (h: P.C.hom E1.1 E2.1) : U
:= Σ (_ : = (P.C.hom E1.1 D.2.1.1)
(P.P.○ E1.1 E2.1 D.2.1.1 h E2.2.1) E1.2.1),
= (P.C.hom E1.1 D.2.2.1)
(P.P.○ E1.1 E2.1 D.2.2.1 h E2.2.2.1) E1.2.2.1

def cospanConeHom (P: precategory) (D: cospan P) (E1 E2: cospanCone P D) : U
:= Σ (h: P.C.hom E1.1 E2.1), isCospanConeHom P D E1 E2 h

def isPullback (P: precategory) (D: cospan P) (E: cospanCone P D) : U
:= Σ (h: cospanCone P D), isContr (cospanConeHom P D h E)

def hasPullback (P: precategory) (D: cospan P) : U
:= Σ (E: cospanCone P D), isPullback P D E

```

1.3 Functor

A **functor** $F : \mathcal{C} \rightarrow \mathcal{D}$ assigns to each:

- Object $X \in \mathcal{C}$ an object $F(X) \in \mathcal{D}$,
- Morphism $f : X \rightarrow Y$ a morphism $F(f) : F(X) \rightarrow F(Y)$,

such that $F(id_X) = id_{F(X)}$ and $F(g \circ f) = F(g) \circ F(f)$.

Definition 4. (Category Functor). Let A and B be precategories. A functor $F : A \rightarrow B$ consists of: (i) A function $F_{Ob} : Ob_A \rightarrow Ob_B$; (ii) for each $a, b : Ob_A$, a function $F_{Hom} : Hom_A(a, b) \rightarrow Hom_B(F_{Ob}(a), F_{Ob}(b))$; (iii) for each $a : Ob_A$, $F_{Ob}(1_a) = 1_{F_{Ob}(a)}$; (iv) for $a, b, c : Ob_A$ and $f : Hom_A(a, b)$ and $g : Hom_A(b, c)$, $F(g \circ f) = F_{Hom}(g) \circ F_{Hom}(f)$.

```
def catfunctor (A B: precategory): U
:= Σ (ob: A.C.ob -> B.C.ob)
      (mor:   Π (x y: A.C.ob) ,
              A.C.hom x y -> B.C.hom (ob x) (ob y))
      (id:    Π (x: A.C.ob) ,
              = (B.C.hom (ob x) (ob x))
                 (mor x x (A.P.id x)) (B.P.id (ob x)))
      (fcomp: Π (x y z: A.C.ob) (f: A.C.hom x y) (g: A.C.hom y z) ,
              = (B.C.hom (ob x) (ob z))
                 (mor x z (A.P.o x y z f g))
                 (B.P.o (ob x) (ob y) (ob z) (mor x y f) (mor y z g))), 1
```

1.4 Terminals

Definition 5. (Terminal Object). Is such object Ob_C , that

$$\prod_{x,y:\text{Ob}_C} \text{isContr}(\text{Hom}_C(y,x)).$$

```
def isInitial (P: precategory) (bot: P.C.ob): U
:= Π (x: P.C.ob), isContr (P.C.hom bot x)

def isTerminal (P: precategory) (top: P.C.ob): U
:= Π (x: P.C.ob), isContr (P.C.hom x top)

def initial (P: precategory): U
:= Σ (bot: P.C.ob), isInitial P bot

def terminal (P: precategory): U
:= Σ (top: P.C.ob), isTerminal P top
```

1.5 Natural Transformation

A **natural transformation** $\eta : F \Rightarrow G$ between functors $F, G : \mathcal{C} \rightarrow \mathcal{D}$ consists of morphisms $\eta_X : F(X) \rightarrow G(X)$ such that for every $f : X \rightarrow Y$ in \mathcal{C} ,

$$\begin{array}{ccc} F(X) & \xrightarrow{\eta_X} & G(X) \\ F(f) \downarrow & & \downarrow G(f) \\ F(Y) & \xrightarrow{\eta_Y} & G(Y) \end{array}$$

commutes.

```
def isNaturalTransformation
  (C D: precategory)
  (F G: catfunctor C D)
  (eta: Π (x: C.C.ob), D.C.hom (F.ob x) (G.ob x)) : U
  := Π (x y: C.C.ob) (h: C.C.hom x y),
  = (D.C.hom (F.ob x) (G.ob y))
  (D.P.○ (F.ob x) (F.ob y) (G.ob y) (F.mor x y h) (eta y))
  (D.P.○ (F.ob x) (G.ob x) (G.ob y) (eta x) (G.mor x y h))

def nattrans (C D: precategory) (F G: catfunctor C D): U
  := Σ (η: Π (x: C.C.ob), D.C.hom (F.ob x) (G.ob x))
  (commute: isNaturalTransformation C D F G η), unit

def natiso (C D: precategory) (F G: catfunctor C D) : U
  := Σ (left: nattrans C D F G)
  (right: nattrans C D G F), 1
```

1.6 Adjunction

An **adjunction** between categories \mathcal{C} and \mathcal{D} consists of functors

$$F : \mathcal{C} \leftrightarrows D : G$$

and natural transformations (unit η and counit ε)

$$\eta : \text{Id}_{\mathcal{C}} \Rightarrow G \circ F, \quad \varepsilon : F \circ G \Rightarrow \text{Id}_{\mathcal{D}}$$

satisfying the triangle identities.

```

ntransL (C D: precategory) (F G: catfunctor C D)
  (f: ntrans C D F G) (B: precategory) (H: catfunctor B C)
  : ntrans B D (compFunctor B C D H F) (compFunctor B C D H G)
= (eta, p) where
  F': catfunctor B D = compFunctor B C D H F
  G': catfunctor B D = compFunctor B C D H G
  eta (x: carrier B): hom D (F'.1 x) (G'.1 x) = f.1 (H.1 x)
  p (x y: carrier B) (h: hom B x y): Path (hom D (F'.1 x) (G'.1 y))
    (compose D (F'.1 x) (F'.1 y) (G'.1 y) (F'.2.1 x y h) (eta y))
    (compose D (F'.1 x) (G'.1 x) (G'.1 y) (eta x) (G'.2.1 x y h))
= f.2 (H.1 x) (H.1 y) (H.2.1 x y h)

ntransR (C D: precategory) (F G: catfunctor C D)
  (f: ntrans C D F G) (E: precategory) (H: catfunctor D E)
  : ntrans C E (compFunctor C D E F H) (compFunctor C D E G H)
= (eta, p) where
  F': catfunctor C E = compFunctor C D E F H
  G': catfunctor C E = compFunctor C D E G H
  eta (x: carrier C): hom E (F'.1 x) (G'.1 x)
= H.2.1 (F.1 x) (G.1 x) (f.1 x)
  p (x y: carrier C) (h: hom C x y): Path (hom E (F'.1 x) (G'.1 y))
    (compose E (F'.1 x) (F'.1 y) (G'.1 y) (F'.2.1 x y h) (eta y))
    (compose E (F'.1 x) (G'.1 x) (G'.1 y) (eta x) (G'.2.1 x y h))
= <i> comp (<,> hom E (F'.1 x) (G'.1 y))
  (H.2.1 (F.1 x) (G.1 y) (f.2 x y h @ i))
  [ (i =
0) -> H.2.2.2 (F.1 x) (F.1 y) (G.1 y) (F.2.1 x y h) (f.1 y),
  (i =
1) -> H.2.2.2 (F.1 x) (G.1 x) (G.1 y) (f.1 x) (G.2.1 x y h) ]

```

```

areAdjoint (C D: precategory)
(F: catfunctor D C)
(G: catfunctor C D)
  (unit: ntrans D D (idFunctor D) (compFunctor D C D F G))
  (counit: ntrans C C (compFunctor C D C G F) (idFunctor C)): U
= prod ((x: carrier C) -> = (hom D (G.1 x) (G.1 x))
          (path D (G.1 x)) (h0 x))
        ((x: carrier D) -> = (hom C (F.1 x) (F.1 x))
          (path C (F.1 x)) (h1 x)) where
  h0 (x: carrier C) : hom D (G.1 x) (G.1 x)
    = compose D (G.1 x) (G.1 (F.1 (G.1 x))) (G.1 x)
    ((ntransL D D (idFunctor D)
      (compFunctor D C D F G) unit C G).1 x)
    ((ntransR C C (compFunctor C D C G F)
      (idFunctor C) counit D G).1 x)
  h1 (x: carrier D) : hom C (F.1 x) (F.1 x)
    = compose C (F.1 x) (F.1 (G.1 (F.1 x))) (F.1 x)
    ((ntransR D D (idFunctor D)
      (compFunctor D C D F G) unit C F).1 x)
    ((ntransL C C (compFunctor C D C G F)
      (idFunctor C) counit D F).1 x)

adjoint (C D: precategory) (F: catfunctor D C) (G: catfunctor C D): U
= (unit: ntrans D D (idFunctor D) (compFunctor D C D F G))
* (counit: ntrans C C (compFunctor C D C G F) (idFunctor C))
* areAdjoint C D F G unit counit

```

1.7 Modification

1.8 The Logic of Cosmos

The **Foundational** 0-layer comprises categories, functors, natural transformations, adjunctions, modifications, and bicategories, where categoriess specify objects and morphisms with associative composition and identities, functors map categories preserving structure, natural transformations define morphisms between functors, adjunctions establish paired functors with unit and counit, modifications extend transformations to 2-categorical contexts, and bicategories introduce 2-morphisms with weak associativity, providing the algebraic framework for categorical spaces.

The **Computational** 1-layer includes locally cartesian closed categories, cartesian model categories, and symmetric monoidal categories, where locally cartesian closed categories equip slice categories with products and exponentials for sequential computations like lambda calculi, cartesian model categories incorporate Quillen model structures with cofibrant terminal objects for homotopical sequential models, and symmetric monoidal categories provide tensor products with symmetry for parallel computations, such as quantum systems, establishing a duality of computational structures.

The **Metatheoretical** 2-layer contrasts fibered categories with model categories, simplicial categories, and simplicial model categories, where fibered categories enable dependent type theories via cartesian morphisms over base categories, model categories define weak equivalences and fibrations for homotopy type theory, simplicial categories enrich over simplicial sets for higher categorical structures, and simplicial model categories combine model structures with simplicial enrichment, distinguishing static, dependent type systems from dynamic, homotopical frameworks.

The **Multidimensional** n -layer encompasses abelian categories, derived categories, categories of spectra, T-spectra, spectral categories, monoidal model categories, AT categories, monoidal relative categories, and symmetric monoidal $(\infty,1)$ -categories, where abelian categories support exact sequences, derived categories localize chain complexes at quasi-isomorphisms, spectra and T-spectra model stable homotopy, spectral categories enrich over spectra, monoidal model and relative categories add homotopical and monoidal structures, AT categories split into pretoposes and abelian categories, and $(\infty,1)$ -categories extend monoidal structures, defining algebraic and geometric dimensions of spaces.

The **Modal** ∞ -layer integrates cohesive topoi, supergeometry, and TED-K theory, where cohesive topoi unify discrete, continuous, and homotopical structures through an adjoint quadruple of functors with modalities, supergeometry equips spaces with $Z/2Z$ -graded sheaves for superspaces, and TED-K theory constructs generalized cohomology in cohesive or supergeometric contexts, providing a comprehensive framework for category and type theorists to model the modal structure of mathematical spaces.